
Abstract

With help of a compact Prolog-based theorem prover for Intuitionistic Propositional Logic, we synthesize,5

given a formula, minimal assumptions under which the formula becomes a theorem.6

After applying our synthesis algorithm to cover basic abductive reasoning mechanisms, we synthesize7

conjunctions of literals that mimic rows of truth tables in classical or intermediate logics and we abduce8

conditional hypotheses that turn theorems in classical or intermediate logics into theorems in intuitionistic9

logic. One step further, we generalize our abductive reasoning mechanism to synthesize more expressive10

sequent premises using a minimal set of canonical formulas, to which arbitrary formulas in the calculus can11

be reduced while preserving their provability.12

Organized as a self-contained literate Prolog program, the paper supports interactive exploration of its13

content and ensures full replicability of our results.14

Keywords: abductive reasoning in intuitionistic logic, theorem synthesis, logic programming and au-15

tomated reasoning, theorem provers for intuitionistic propositional logic, implementing sequent calculi in16

Prolog.17

1 Introduction18

Given a formula F in Classical Propositional Logic (CL), each row in a formula’s truth table19

describes a conjunction of literals C. Reading the truth table as a disjunctive normal form, it20

immediately follows that C→ F is a tautology. As an example, let us consider the CL formula21

F = (A v B) & (B v C) & (C v A). Then its truth table (with 1 for True and 0 for False) is22

the one on the left. Let us select any row, say [1,0,1] and interpret it as G = A & ~B & C. Then23

the truth table of the resulting tautology G -> F is shown on the right.24

25

A B C : F A B C: G->F26

[0,0,0]-->0 [0,0,0]-->127

[0,0,1]-->0 [0,0,1]-->128

[0,1,0]-->0 [0,1,0]-->129

[0,1,1]-->1 [0,1,1]-->130

[1,0,0]-->0 [1,0,0]-->131

[1,0,1]-->1 <= selected row [1,0,1]-->132

[1,1,0]-->1 [1,1,0]-->133

[1,1,1]-->1 [1,1,1]-->134
35

Thus, it is easy to express, for a formula F in CL, what assumptions would make it a theorem36

in CL.37

This model-theoretic approach extends also to intermediate logics1 and in particular, it applies38

to the 5-valued truth-tables of the equilibrium logic (Pearce et al. 2000), underlying Answer Set39

Programming (ASP).40

With no finite truth-tables, no inter-definability of logical connectives, no rule of excluded41

middle and only a concept of tautology and contradiction defined for Intuitionistic Propositional42

Logic (IL), we need to be a bit more creative when trying to find salient assumptions that would43

make the formula a theorem in IL. First, given a formula in IL, we will need a search process for44

finding assumptions that would make it a theorem. Next, we would like our assumptions to be45

minimal with respect to the partial order relation governing the logic (or its equivalent Heyting46

algebra) , intuitionistic implication.47

This brings us to Abductive Logic Programming (Eshghi and Kowalski 1989; Denecker and48

Kakas 2002), where facts designated as abducibles are filtered with integrity constraints to pro-49

vide relevant assumptions needed for the success of a goal G w.r.t. a given program P. In the50

context of IL, our abductive reasoning will rely on finding minimal assumptions under which a51

formula becomes a theorem.52

And finally, in the absence of a convenient automated semantic method like truth tables or53

SAT solvers in CL, we will need a theorem prover, ideally derived directly from the rules of a54

terminating sequent calculus, that interoperates smoothly with the search process synthesizing55

our assumptions.56

These requirements make Prolog a natural meta-language for an actionable description of57

these concepts. We will materialize our approach as a literate Prolog program, from which, as a58

convenience to the reader, we will extract our code and make it available online as a Prolog file2.59

The rest of the paper is organized as follows. Section 2 overviews our Prolog-based theorem60

prover and the sequent calculus it is derived from. Section 3 introduces our protasis synthesizer61

and its uses for abductive reasoning. Section 4 generalizes our approach to the synthesis of mini-62

mal canonical assumptions. Section 5 discusses significance of our results, its possible extensions63

as well as some of its limitations. Section 6 overviews related work and section 7 concludes the64

paper.65

We assume the reader is fluent in Prolog, propositional intuitionistic and classical logic and fa-66

miliar with abductive reasoning and key concepts behind sequent calculi and automated theorem67

proving.68

2 Background: The Intuitionistic Propositional Logic Theorem Prover69

We will derive our Prolog prover from a set of compact and elegant sequent calculus rules for-70

mally describing provability in IL.71

2.1 Roy Dyckhoff’s G4ip calculus72

Motivated by problems related to loop avoidance in implementing Gentzen’s LJ calculus, Roy73

Dyckhoff designed and proved sound and complete a sequent calculus-based axiomatization of74

IL (Dyckhoff 1992) . He has proved that the calculus is terminating, by identifying a multiset75

ordering-based formula size definition that decreases after each step (Dyckhoff 1992).76

1 logics weaker than classical but stronger than intuitionistic
2 at https://github.com/ptarau/TypesAndProofs/blob/master/isynt.pro

The sequents of the G4ip calculus follow:77

Γ, p⇒ p Ax (p an atom) Γ,⊥⇒ ∆ L⊥

Γ⇒ ϕ Γ⇒ ψ

Γ⇒ ϕ ∧ψ
R∧

Γ,ϕ,ψ ⇒ ∆

Γ,ϕ ∧ψ ⇒ ∆
L∧

Γ⇒ ϕi

Γ⇒ ϕ0∨ϕ1
R∨ (i = 0,1)

Γ,ϕ ⇒ ∆ Γ,ψ ⇒ ∆

Γ,ϕ ∨ψ ⇒ ∆
L∨

Γ,ϕ ⇒ ψ

Γ⇒ ϕ → ψ
R→

Γ, p,ϕ ⇒ ∆

Γ, p, p→ ϕ ⇒ ∆
Lp→ (p an atom)

Γ,ϕ → (ψ → γ)⇒ ∆

Γ,ϕ ∧ψ → γ ⇒ ∆
L∧→

Γ,ϕ → γ,ψ → γ ⇒ ∆

Γ,ϕ ∨ψ → γ ⇒ ∆
L∨→

Γ,ψ → γ ⇒ ϕ → ψ γ,Γ⇒ ∆

Γ,(ϕ → ψ)→ γ ⇒ ∆
L→→

Key to the termination proof in (Dyckhoff 1992) is the rule L→→ that breaks down nested78

implications into “smaller” ones, each containing fewer connectives. The rules work with the79

context Γ being a multiset, but it has been shown later (Dyckhoff 2016) that Γ can be a set, with80

duplication in contexts eliminated.81

Note that the same calculus has been discovered independently in the 50’s by Vorob’ev and in82

the 80’s-90’s by Hudelmaier (Hudelmaier 1988).83

2.2 Implementing the Theorem Prover84

In the tradition of ”lean theorem provers”, we can build one directly from the G4ip calculus, in85

a goal oriented style, by reading the rules from conclusions to premises.86

Thus, we start with a simple, almost literal translation of sequent rules to Prolog with values87

in the environment Γ denoted by the variable Vs. Besides implication (denoted ->), conjunction88

(denoted &) and disjunction (denoted v), we implement rules for inverse implication (denoted89

<-), negation (denoted ~) and equivalence (denoted <->). We also add rules for a top element90

(denoted true) and a bottom element (denoted false).91

Correctness of our additions follows from the definitions of:92

• intuitionistic inverse implication: ϕ ← ψ ≡ ψ → ϕ93

• intuitionistic equivalence: ϕ ↔ ψ ≡ ϕ → ψ & ψ → ϕ94

• intuitionistic negation: ~ϕ ≡ ϕ → false.95

• top element: true ≡ false→ false.96

Note that the added connectives are meant to enhance the expressiveness of the logic. For in-97

stance, “<->” allows expressing the fact that two formulas are equivalent and thus equiprovable,98

“head <- body” mimics Prolog’s familiar Horn clause syntax “head :- body” and finally the99

negation symbol makes formulas more compact and human-readable.100

We define operators for all our connectives, except ->, that we will use with its standard right101

associativity.102

103

:- op(525, fy, ~ ).104

:- op(550, xfy, & ). % right associative105

:- op(575, xfy, v ). % right associative106

:- op(600, xfx, <-> ). % non associative107

:- op(800, yfx, <- ). % left associative108
109

A formula T is a theorem if it is provable from an empty set of assumptions:110

111

iprover(T) :- iprover(T,[]).112
113

We follow here Dyckhoff’s calculus but delegate details to helper predicates iprover reduce/4114

and iprover impl/4.115

116

iprover(true,_):-!.117

iprover(A,Vs):-memberchk(A,Vs),!.118

iprover(_,Vs):-memberchk(false,Vs),!.119

iprover(~A,Vs):-!,iprover(false,[A|Vs]).120

iprover(A<->B,Vs):-!,iprover(B,[A|Vs]),iprover(A,[B|Vs]).121

iprover((A->B),Vs):-!,iprover(B,[A|Vs]).122

iprover((B<-A),Vs):-!,iprover(B,[A|Vs]).123

iprover(A & B,Vs):-!,iprover(A,Vs),iprover(B,Vs).124

iprover(G,Vs1):- % atomic or disj or false125

select(Red,Vs1,Vs2),126

iprover_reduce(Red,G,Vs2,Vs3),127

!,128

iprover(G,Vs3).129

iprover(A v B, Vs):-(iprover(A,Vs) ; iprover(B,Vs)),!.130
131

iprover reduce/4 is the first step in breaking down formulas in the premise into their compo-132

nents.133

134

iprover_reduce(true,_,Vs1,Vs2):-!,iprover_impl(false,false,Vs1,Vs2).135

iprover_reduce(~A,_,Vs1,Vs2):-!,iprover_impl(A,false,Vs1,Vs2).136

iprover_reduce((A->B),_,Vs1,Vs2):-!,iprover_impl(A,B,Vs1,Vs2).137

iprover_reduce((B<-A),_,Vs1,Vs2):-!,iprover_impl(A,B,Vs1,Vs2).138

iprover_reduce((A & B),_,Vs,[A,B|Vs]):-!.139

iprover_reduce((A<->B),_,Vs,[(A->B),(B->A)|Vs]):-!.140

iprover_reduce((A v B),G,Vs,[B|Vs]):-iprover(G,[A|Vs]).141
142

iprover impl/4 details the case analysis of the handling of implication (and its instances)143

prescribed by rule L→→.144

145

iprover_impl(true,B,Vs,[B|Vs]):-!.146

iprover_impl(~C,B,Vs,[B|Vs]):-!,iprover((C->false),Vs).147

iprover_impl((C->D),B,Vs,[B|Vs]):-!,iprover((C->D),[(D->B)|Vs]).148

iprover_impl((D<-C),B,Vs,[B|Vs]):-!,iprover((C->D),[(D->B)|Vs]).149

iprover_impl((C & D),B,Vs,[(C->(D->B))|Vs]):-!.150

iprover_impl((C v D),B,Vs,[(C->B),(D->B)|Vs]):-!.151

iprover_impl((C<->D),B,Vs,[((C->D)->((D->C)->B))|Vs]):-!.152

iprover_impl(A,B,Vs,[B|Vs]):-memberchk(A,Vs).153
154

Example 1155

After defining:156

157

iprover_test:-158

Taut = ((p & q) <-> (((p v q)<->q)<->p)), iprover(Taut),159

Contr=(a & ~a), \+ (iprover(Contr)).160
161

we observe success on proving a tautology and failing to prove a contradiction, but we refer to162

(Tarau 2019) for an extensive combinatorial testing of a variant of this prover as well as its testing163

against the ILTP benchmark3 and several other provers.164

2.3 Classical Logic For Free165

Glivenko’s theorem states that a propositional formula F is a classical tautology if and only if166

˜ ˜F is an intuitionistic tautology. This gives us a classical prover for free, that we will use as an167

alternative to iprove when defining several concepts parameterized by a prover.168

169

% classical prover - via Glivenko's theorem170

cprover(T):-iprover( ~ ~T).171
172

Example 2173

The two provers, as it is well known, will disagree on p v ~p174

175

?- iprover(p v ~p).176

false.177

?- cprover(p v ~p).178

true.179
180

but agree on p & ~p:181

182

?-iprover(p & ~p).183

false.184

?-cprover(p & ~p).185

false.186
187

3 Abductive Reasoning Mechanisms188

We are now ready to introduce our search for assumptions that make a given formula an intu-189

itionistic tautology.190

3.1 Generating the Abducibles191

Defining some of the atoms occurring in a formula F as the only ones to be used in the search pro-192

cess brings us to declare them as abducibles (Eshghi and Kowalski 1989), but we will also enable193

the option to make abducible all the atoms occurring in F . Thus, when the variable Abducibles194

is free, all atomic symbols will be considered abducibles.195

196

abducibles_of(Formula,Abducibles):-var(Abducibles),!,atoms_of(Formula,Abducibles).197

abducibles_of(_,_).198
199

3 http://www.iltp.de/

The predicate atoms of/2 finds the set of all the atoms occurring in a formula. It backtracks200

over all arguments, recursively and it collects atomic elements to a list, with setof/3 which also201

eliminates possible duplicates.202

203

atom_of(A,R):-atomic(A),!,R=A.204

atom_of(T,A):-arg(_,T,X),atom_of(X,A).205

206

atoms_of(T,As):-setof(A,atom_of(T,A),As).207
208

3.2 Protasis Generation209

If F is a formula, we can think of a premise C such that C→ F is a theorem4 as a counterfac-210

tual assumption making F conditionally true. We call such a formula C a protasis. Thus, given211

a formula and a set of assumptions (our abducibles), we will need to search among them for212

assumptions that would make the formula a theorem.213

The predicate any protasis/6 implements this idea, subject to a set of parameters:214

• Prover allows a choice of the underlying logic (e.g., intuitionistic or classical)215

• AggregatorOp fixes one of the connectives of the logic, from which the protasis is built 5
216

• The yes/no flag WithNeg decides if negations of the abducibles can be part of the protasis217

• Abducibles is a set of atoms or a free variable, meaning that all atoms will be included218

• Assumption will be any protasis, that given the previously specified parameters, ensures219

that Assumption->Formula is a theorem in the logic specified by Prover.220

221

any_protasis(Prover,AggregatorOp,WithNeg,Abducibles,Formula,Assumption):-222

abducibles_of(Formula,Abducibles),223

mark_hypos(WithNeg,Abducibles,Literals),224

subset_of(Literals,Hypos),225

join_with(AggregatorOp,Hypos,Assumption),226

\+ (call(Prover,Assumption->false)), % we do not assume contradictions !227

call(Prover,Assumption->Formula). % we ensure this is a theorem228
229

The predicate mark hypos/3 will mark with their negations the abducibles if we want to allow230

them to occur in the protasis positively or prefixed by their negations.231

232

mark_hypos(_,[],[]).233

mark_hypos(yes,[P|Ps],[P,~P|Ns]):-mark_hypos(yes,Ps,Ns).234

mark_hypos(no,[P|Ps],[P|Ns]):-mark_hypos(no,Ps,Ns).235
236

Our subset generator subset of, to be used to iterate over all subsets of the abducibles, first237

enumerates templates of increasing length as we want smaller subsets to be tried first.238

239

subset_of(Xs,Ts):-template_from(Xs,Ts),tsubset(Xs,Ts).240

241

template_from(_,[]).242

template_from([_|Xs],[_|Zs]):-template_from(Xs,Zs).243
244

4 or, equivalently, when C is the premise and F is the conclusion of a provable sequent
5 the restriction to on operator will be lifted later in section4, when we generalize this mechanism to a set of canonical

formulas

Then, for each template of length K, it fills it with a subset of length K of the N abducibles, one245

at a time, on backtracking.246

247

tsubset([],[]).248

tsubset([X|Xs],[X|Rs]):-tsubset(Xs,Rs).249

tsubset([_|Xs],Rs):-tsubset(Xs,Rs).250
251

The predicate join with op builds an expression from a sequence of abducible literals (atoms,252

possibly negated) with a given operator.253

254

join_with_op(_,[],true).255

join_with_op(_,[X],X).256

join_with_op(Op,[X,Y|Xs],R):-join_with_op(Op,[Y|Xs],R0),R=..[Op,X,R0].257
258

How we join the abducible literals with help of a given operator, is different for associative and259

commutative operators like & and v (the default 3-rd clause of join with) and ->, <-, <->, that260

we treat as special cases.261

Thus, once the head was picked with select/3, the order of the remaining literals is immate-262

rial.263

264

join_with(Op,Xs,R):-265

memberchk(Op,[(->),(<-)]),!,266

select(Head,Xs,Ys), append(Ys,[Head],Zs),267

join_with_op((->),Zs,R).268
269

For non-associative <-> we will use all permutations of the abducibles.270

271

join_with(Op,Xs,R):-Op=(<->),!,permutation(Xs,Ys),join_with_op(Op,Ys,R).272
273

Finally, as v and & are permutation invariant, we just call join with op.274

275

join_with(Op,Xs,R):- join_with_op(Op,Xs,R).276
277

Implication and reverse implications are handled in join with, knowing that their components,
except the head, are permutation invariant, with the following equivalences in mind:

ϕ0← ϕ1 · · · ← ϕn ≡ ϕ0← ϕ1 & . . . & ϕn

and

ϕn→ ϕn−1 · · · → ϕ1→ ϕ0 ≡ ϕn & ϕn−1 & . . . & ϕ1 → ϕ0

Note that these hold both intuitionistically and classically, as it can be quickly verified with278

iprover and cprover.279

3.3 The Weakest Protasis280

The next step is defining a weakest protasis, keeping in mind that a partial order6 among them is281

defined by the intuitionistic implication (->). First, we will collect with setof/3 all the candidate282

assumptions to ensure that the prover is called on each only once.283

6 in contrast to classical logic, where (p->q) v (q->p) is a theorem

284

weakest_protasis(Prover,AggregatorOp,WithNeg,Abducibles,Formula,Assumption):-285

setof(Assumption,286

any_protasis(Prover,AggregatorOp,WithNeg,Abducibles,Formula,Assumption),287

Assumptions),288

weakest_with(Prover,Assumptions,Assumption).289
290

Next, we ensure that a weakest protasis is such that it does not imply any other protasis, thus that291

it is a minimal element w.r.t. our partial order. We rely for that on the predicate weakest with/3:292

293

weakest_with(_,Gs,G):-memberchk(true,Gs),!,G=true.294

weakest_with(Prover,Gs,G):-select(G,Gs,Others),295

\+ (member(Other,Others),weaker_with(Prover,Other,G)).296
297

The partial order relation is exposed as the predicate weaker with/3 which ensures that a weak-298

est protasis does not imply any other protasis, including ones that might imply it.299

300

weaker_with(Prover,P,Q):- \+ call(Prover,(P->Q)), call(Prover,(Q->P)).301
302

Note that the predicate weakest protasis/6 depends on the same parameters as any protasis,303

in particular on the Prover implementing a given provability relation.304

Example 3305

Peirce’s law is known to hold in CL and not hold in IL. In fact, it can turn IL into CL if added306

as an axiom. The predicate peirce/2 will try to synthesize an assumption that would make it307

hold.308

309

peirce(Prover,WhatIf):-310

Formula=(((p->q)->p)->p),311

WithNeg=yes, AggregatorOp=(v), Abducibles=[p],312

weakest_protasis(Prover,AggregatorOp,WithNeg,Abducibles,Formula,WhatIf).313
314

When running it we get:315

316

?- peirce(iprover,Protasis).317

Protasis = p v ~p.318

?- peirce(cprover,Protasis).319

Protasis = true.320
321

This reveals an interesting fact. It tells us that if p v ~p were assumed, Peirce’s law would hold322

in IL, as iprove would succeed. As it is well known, p v ~p would turn IL into CL and cprove323

tells us that indeed, Peirce’s law holds unconditionally in CL.324

Example 4325

We can also synthesize conditional assumptions when using implication or inverse implication326

as our aggregator connective. After defining:327

328

impl_aggr(H):-329

T=(a<-((a<-(b<-d))&(b<-c))),330

Prover=iprover, WithNeg=yes, AggregatorOp=(->), As=[c,d],331

weakest_protasis(Prover,AggregatorOp,WithNeg,As,T,H).332
333

we obtain:334

335

?- impl_aggr(H).336

H = ( d->c).337
338

showing that the implication d->c (equivalent of the Horn Clause c:-d) should hold for the339

formula to be a theorem. This illustrates a mechanism to synthesize Horn Clauses playing the340

role of conditional assumptions.341

Another interesting case is that of a contradiction. After defining contra test/1 as:342

343

contra_test(H):-344

T=(p & ~p),345

Prover=iprover, WithNeg=yes, AggregatorOp=(&),346

weakest_protasis(Prover,AggregatorOp,WithNeg,_Abducibles,T,H).347
348

and running it with:349

350

?- contra_test(Protasis).351

false.352
353

we can see that no assumption would make the contradiction a tautology, and this will also be354

the case for Prover=cprover. Note that to ensure this, we have enforced in the definition of355

any protasis that such assumptions should themselves not be contradictions.356

Example 5357

In the case of the logic of here-and-there (Pearce 1997), derived from IL by adding the axiom358

f v (f->g) v ~g359

as shown in (Lifschitz et al. 2001), we will get with cprover the protasis true. This indicates, as360

expected, that it is already a theorem in CL and thus also a theorem in the logic of here-and-there.361

362

?- weakest_protasis(cprover,(v),yes,_,(f v (f->g) v ~g),P).363

P = true.364
365

On the other hand, the less obvious weakest protasis obtained for iprove indicates that the366

excluded middle rule would be needed for both f and g.367

368

?- weakest_protasis(iprover,(v),yes,_,(f v (f->g) v ~g),P).369

P = f v ~f v g v ~g.370
371

3.4 An Example of Intuitionistic Abductive Reasoning372

With the logic of synthesizing meaningful minimal assumptions that make a formula a theorem373

clarified, we are now ready to revisit abductive reasoning along the lines of (Eshghi and Kowalski374

1989).375

The predicate explain with/5 finds, given a Prover, the abductive inference problem pa-376

rameterized by:377

• a formula Prog seen here as representing a knowledge base378

• a set of Abducibles occurring in Prog379

• a goal formula G such that Prog -> G should always hold380

• a formula IC playing the role of integrity constraints meant to filter out unwanted assump-381

tions382

383

explain_with(Prover,Abducibles,Prog,IC,G,Expl):-384

any_protasis(Prover,(&),yes,Abducibles,(Prog->G), Expl),385

call(Prover, Expl & Prog->G),386

call(Prover,(Expl & Prog->IC)),387

\+ (call(Prover,(Expl & Prog -> false))).388
389

Note also that any protasis replaces weakest protasis in this definition, given that mini-390

mality might want to be stated as part of the integrity constraints IC.391

Example 6392

To revisit a simple example of abductive explanation generation, we define:393

394

why_wet(Prover):-395

IC = ~(rained & sunny),396

P = sunny & (rained v sprinkler -> wet), As=[sprinkler,rained], G = wet,397

writeln(prog=P), writeln(ic=IC),398

explain_with(Prover,As,P,IC,G,Explanation),399

writeln('Explanation:' --> Explanation).400
401

Then, when running it, it will display, as expected:402

403

?- why_wet(iprover).404

prog=sunny&(rained v sprinkler->wet)405

ic= ~ (rained&sunny)406

Explanation: --> sprinkler& ~rained407
408

4 Synthesis of Minimal Canonical Assumptions409

We will now generalize our abductive reasoning mechanism by lifting the constraint on the410

premise of our sequent from literals connected by a single operation to a canonical form that411

has been shown to be able to represent arbitrary IL formulas.412

4.1 The Mints Transformation413

Grigori Mints has proven, in his seminal paper studying complexity classes for intuitionistic
propositional logic (Mints 1992), that any formula f is equiprovable to a formula of the form
X f → g where X f is a conjunction of formulas of one of the forms:

p, ˜p, p→ q, (p→ q)→ r, p→ (q→ r), p→ (q v r), p→ ˜q, ˜q→ p.

With introduction of new variables (like with the Tseitin transform for SAT or ASP solvers), the414

transformation runs in linear space and time. Note that as a premise to a sequent can be seen as a415

conjunction implying its conclusion, the conjunction of the formulas described by Mints can be416

seen as serving the same purpose as the conjunctive normal form (CNF) in classical logic.417

Thus, by generating this set of bounded size formulas as premises of a sequent, we can express418

equivalent formulas of unbounded size, otherwise subject to a much larger search space in the419

formula synthesis process.420

We will now generate, using a given set of abducibles, premises built of these formulas, with421

their propositional variables selected from the set of abducibles.422

Conceptually, while we will keep calling the propositional variables involved in our premise423

abducibles, we should see them from now on simply as a set of independent variables on which424

the derivation of the sequent’s conclusion from its premise depends.425

We define a generator for the set of Mints formulas with help of a DCG grammar7 that collects426

its propositional variables from a list of abducibles.427

428

mints_formula(P)-->[P]. mints_formula(~P)-->[P].429

mints_formula((P->Q))-->[P,Q]. mints_formula((P->Q)->R)-->[P,Q,R].430

mints_formula((P->(Q->R)))-->[P,Q,R]. mints_formula((P->(Q v R))) -->[P,Q,R].431

mints_formula((P-> ~Q))-->[P,Q]. mints_formula((~P->Q))-->[P,Q].432
433

We extend our DCG, subject to the same limitation on available abducibles, to generate a list434

of formulas meant to be part of the premise. We will later aggregate this list into a conjunction.435

436

mints_conjuncts([])-->[].437

mints_conjuncts([F|Fs])-->mints_formula(F),mints_conjuncts(Fs).438
439

Next, we eliminate duplicates with Prolog’s sort/2.440

441

mints_conjuncts(Atoms,Conjuncts):-mints_conjuncts(Ps,Atoms,[]),sort(Ps,Conjuncts).442
443

We derive any mints premise/4 in a way similar to any protasis/6, except that the ar-444

guments WithNeg and AggregatorOp become unnecessary and multiple occurrences of any445

abducible need to be supported. For brevity, we explain our steps as comments in the code.446

447

any_mints_premise(Prover,Abducibles,Formula,Premise):-448

abducibles_of(Formula,Abducibles),449

subset_of(Abducibles,Chosen), % select a subset of Abducibles450

template_from(Abducibles,Atoms), % Atoms is a list of free variables451

part_as_equiv(Atoms,Chosen), % Chosen provides unique occurrences of Atoms452

mints_conjuncts(Atoms,Conjuncts), % builds the Mints formulas453

join_with_op((&),Conjuncts,Premise), % joins Conjuncts into a conjunction454

\+ (call(Prover,Premise->false)), % ensures Premise is not a contradiction455

call(Prover,Premise->Formula). % ensure that Premise implies Formula456
457

Note that we have limited the length of the premise in the case of the Mints-formulas, arbitrarily,458

to the number of abducible atoms, that the selection of Atoms has as an upper bound. For more459

flexibility, this can be lifted to be based on a length parameter passed to any mints premise.460

4.2 Labeling the Variables in the Mints Formulas461

We will describe here the implementation of part as equiv/2 that will be used to to gener-462

ate variables bound to possibly repeated occurrences of each atom. Note that equalities of logic463

variables define equivalence classes that correspond to partitions of the set of variables. We im-464

plement this simply by selectively unifying them.465

The predicate part as equiv/2 takes a list of distinct logic variables and generates partitions-466

as-equivalence-relations by unifying them “nondeterministically”. It also collects the unique467

variables defining the equivalence classes, as a list given by its second argument. It works re-468

versibly, when unique values are given as its second argument and a bound list of free variables469

as its first.470

7 As a note to the reader unfamiliar with Prolog’s Definite Clause Grammars (DCG) preprocessor, it transforms a DCG
clause like a --> b,c,d into an ordinary Prolog clause a(S0,Sn) :- b(S0,S1),c(S1,S2),d(S2,Sn), to con-
veniently keep track of state changes in the “chained” variables S0,S1,...,Sn.

471

part_as_equiv([],[]).472

part_as_equiv([U|Xs],[U|Us]):-complement_of(U,Xs,Rs),part_as_equiv(Rs,Us).473
474

To implement it, we split a set repeatedly in subset+complement pairs with help from the predi-475

cate complement of/2.476

477

complement_of(_,[],[]).478

complement_of(U,[X|Xs],NewZs):-complement_of(U,Xs,Zs),place_element(U,X,Zs,NewZs).479

480

place_element(U,U,Zs,Zs).481

place_element(_,X,Zs,[X|Zs]).482
483

Example 7484

Here, we are interested in the reverse use of part as equiv, with the list of unique variables as485

input and a sequence of variables of fixed length but possibly repeated occurrences as output.486

487

?- length(Vs,4),part_as_equiv(Vs,[a,b]).488

Vs = [a, b, a, a] ; Vs = [a, a, b, a] ; Vs = [a, b, b, a] ;489

Vs = [a, a, a, b] ; Vs = [a, b, a, b] ; Vs = [a, a, b, b] ; Vs = [a, b, b, b] .490
491

We derive weakest mints premise/4 in a way similar to weakest protasis/6 except for492

passing only the relevant arguments to any mints premise/4.493

494

weakest_mints_premise(Prover,Abducibles,Formula,Premise):-495

setof(Premise,496

any_mints_premise(Prover,Abducibles,Formula,Premise),497

Premises),498

weakest_with(Prover,Premises,Premise).499
500

Example 8501

When using the axiom that conservatively extends IL to the logic of here-and-there (Lifschitz502

et al. 2001) we observe again that no premise is needed for cprover and that iprover suggests503

as premises either one of the disjuncts in the axiom, or, more interestingly, g-> ~g.504

505

?- weakest_mints_premise(cprover,_,(f v (f->g) v ~g),P).506

P = true.507

?- weakest_mints_premise(iprover,_,(f v (f->g) v ~g),P).508

P = f ; P = ~g ; P = ( f->g) ; P = ( g-> ~g).509
510

In fact, we observe:511

512

?- iprover((g -> ~g) <-> ~g).513

true.514
515

suggesting that extending IL with:516

f v (f->g) v (g -> ~g)517

would result in an alternative axiomatization of the logic of here-and-there.518

5 Discussion519

We hope that the astute reader is aware at this point that the paper is an exploration of the theory520

behind some fundamental concepts relating abductive reasoning, program synthesis and theorem521

proving in a concise and easily replicable form, facilitated by the choice of Prolog as our meta-522

language, but with the possibility of a fairly routine transliteration to a traditional “formulas-on-523

paper” presentation in mind. As such, the paper can be seen as an executable specification of524

these concepts. While not neglecting minimal efforts for efficient execution, and some elegance525

in the coding style, our main priority was to ensure that the paper conveys its message as a fully526

self-contained literate program.527

We have designed our abductive reasoning logic entirely in a proof-theoretical framework, in528

contrast to the usual model-theoretical semantics, arguably in the original spirit of intuitionistic529

logic.530

Both our weakest protasis-based and weakest Mints formulas-based synthetic assumptions531

are attempts to recover in IL an analogue of the CNF available for a formula in CL. At the532

same time, finding the weakest assumptions shares the focus on minimal models encountered533

in various logic calculi. Our interest in finding weakest assumptions under which the formula534

becomes a theorem is driven by the transitivity of the partial order induced by ->, given that for535

a given formula f, becoming a theorem under the assumption w, ensures also that if (s->w),536

then (s->f) is also a theorem, where w denotes a weakest assumption and s denotes a (stronger)537

assumption that implies it.538

We have restricted ourselves to propositional logic but we foresee extensions to stronger logics539

among which Monadic First Order Logic (known as decidable for CL and undecidable for IL),540

enhanced with Prolog’s constraint solving mechanisms is a promising option.541

While we have forced a clear separation between our meta-language (Prolog) and object-542

language (IL), it would be quite easy to extend our theorem prover to reflect Prolog’s negation as543

failure in the object-language as an addition to IL. This would result in a logic with two flavors544

of negation, similar in the context of IL to the underlying equilibrium logic of ASP.545

6 Related work546

We refer to (Denecker and Kakas 2002) as still the most lucid and comprehensive overview (also547

citing 124 papers) on abductive reasoning in Logic Programming and to (Eshghi and Kowal-548

ski 1989) as one of the most influential initiators for the interest in the field, with connections549

explored in depth to negation as failure and non-monotonic reasoning. Some of our examples550

related to the logic of here-and-there and equilibrium logic (Pearce 1996; Pearce et al. 2000)551

originate in (Lifschitz et al. 2001), where intermediate logics relevant for the foundation of ASP552

systems are overviewed. For abductive reasoning in the context of several logics including non-553

monotonic ones, we mention (Gabbay and Olivetti 2002) and (Gabbay 2000).554

By contrast, the novelty of our approach is the generalized view of abductive reasoning as555

an instance of program synthesis controlled by a theorem prover. Our theorem prover is derived556

directly from the G4ip sequent calculus (Dyckhoff 1992; Dyckhoff 2016). In (Tarau 2019) details557

of this derivation process as well as a combinatorial testing framework used to insure correctness558

are given. The idea of using a theorem-prover for the synthesis of modal formulas is also present559

in (Tarau 2020), having as an outcome an embedding of the epistemic logic IEL in IL and a560

derived theorem prover for that logic. In (Tarau and de Paiva 2020) a theorem prover, restricted561

to the implicational fragment of IL is used to derive a theorem prover for implicational linear562

logic, with help from the Curry-Howard correspondence and the use of linearity of the resulting563

lambda terms as a filtering mechanism. In this context, the distinct focus of the current paper is564

on a very general formula synthesis mechanism within IL itself, covering abductive reasoning565

and emulating in IL key semantic concepts available in CL and intermediate logics.566

7 Conclusions567

We have presented a fully executable specification of a generalized abductive reasoning frame-568

work, that can be relatively easily ported to any logic for which a decision mechanism exists569

(e.g., as provided by a theorem prover). In particular, this applies to several interesting interme-570

diate logics among which the equilibrium-logic (relevant as a foundation of ASP systems) as571

well as modal logics and their instantiations as alethic, deontic or epistemic systems. Besides572

providing (in the form of the concept of weakest protasis) an analogue of the unavailable truth-573

table models for intuitionistic formulas, we have also generalized our abduced sequent premises574

to use minimal canonical formulas to which arbitrary IL formulas can be broken down, with the575

potential of synthesizing salient assumptions that would make a given formula a theorem. When576

the underlying logic is used to model a set of safety constraints that should always hold, this577

generalized abduction synthesis could reveal critical missing assumptions, not just as literals but578

also as a conjunction of interdependencies among them.579

Addressing Reviewers Comments624

Warm thanks to the reviewers for their careful reading of the paper, their constructive suggestions625

and detailed comments!626

Reviewer 1627

>>> 1-2 I am not sure that the title is correct. Does the paper actually present an abductive logic?628

I do not think so. Please consider reformulating629

Changed title, to clarify that it is about abductive reasoning mechanisms rather than a new630

logic. It is now:631

Abductive Reasoning in Intuitionistic Propositional Logic via Theorem Synthesis632

>>> l.5-17 The abstract is rather obscure. Instead of the current text, a high-level explanation of633

key results would do a better job634

Clarified the abstract with emphasis on contributions.635

>>> l.86 ? 90 Additional connectives are introduced, without any explanation of why they are636

really needed. If you wanted to save space and make the code lighter, you could give just basic637

connectives in the paper, referring the reader to explore the rest in the code itself. Even in the later638

case, I?d like an explanation of why those additional connectives are particularly interesting in639

this context640

Added the following justification for the extra connectives:641

Note that the added connectives are meant to enhance the expressiveness of the logic. For instance,642

“<->” allows expressing the fact that two formulas are equivalent and thus equiprovable, “head <- body”643

mimics Prolog’s familiar Horn clause syntax “head :- body” and finally the negation symbol makes644

formulas more compact and human-readable.645

>>> l.211 AggregatorOp for some reason allows only 1 connective to be chosen. This comes out646

of the blue in this part of text. Firstly, it is a serious limitation, and could be declared earlier.647

Secondly, the reason for the restriction is not given. I assume it is to cut down the search space?648

Whatever the reason is, it is worth explaining. added a footnote explaining that:649

the restriction to on operator will be lifted later in section4, when we generalize this mechanism to a set650

of canonical formulas651

>>> l. 368: the section?s title (Section 3.4) is a bit misleading. It is really just a wrapper code652

and an example, whereas the title suggests this is a section about implementation. Think about653

replacing please. Rephrased subsection title as:654

An Example of Intuitionistic Abductive Reasoning655

>>> . 409 ? 410: a formula f is equiprovable ? I think you mean ?any formula f in IFL is656

equiprovable?657

Applied change of “formula” to “any formula”.658



Abductive Reasoning in Intuitionistic Propositional Logic via Theorem Synthesis 17

>>> l. 424 ? 435: please explain the meaning of the syntax “-->”659

Added as footnote:660

As a note to the reader unfamiliar with Prolog’s Definite Clause Grammars (DCG) preprocessor, it trans-661

forms a DCG rule like a --> b,c,d into an ordinary Prolog clause a(S0,Sn) :- b(S0,S1),c(S1,S2),d(S2,Sn),662

to conveniently keep track of state changes in the “chained” variables S0,S1,...,Sn.663

Reviewer 2664

>>> When discussing extensions to the work, I question whether the specific techniques being665

employed can still apply: e.g. most likely the findall/3 calls would misbehave in the presence of666

constraints or compound terms with variables - a more robust mechanism will be called for.667

We have replaced in the code calls to findall/3 with equivalent calls to setof/3 which could also668

handle terms with variables in extensions requiring it.669

>>> detail: p.2: what is a ”salient” assumption?670

>>> x671

Reviewer 3672

>>> Pg 1 - Thus, it is easy to find, for a formula F in CL, a set of assumptions which make it a673

theorem36 in CL. - Thought that task was NP-complete?674

To avoid ambiguity of “easy to find” (which was not a claim about efficiency) , rephrased as:675

Thus, it is easy to express, for a formula F in CL, what assumptions would make it a theorem in CL.676


